Aero Group A2-2 Odometry Code

Odometry.ino - Arduino logic

```
/*
 *      Odometry Task - Aero 2 Group 2
 *
 *          Code written by Duncan R Hamill - 28262174
 *          Tested & verified by Tom Griffiths - 28290771, Ali Hajizadah - 29053056, Robin Hannaford -
 *
 *          All distances in mm, all angles in degrees
 */

#include "defines.h"
#include "encoderInteraction.cpp"
#include "Course.cpp"

#include <Wire.h>
#include <Servo.h>

void setup() {
    // start serial for monitoring
    Serial.begin(9600);

    // setup the I2C and wait 100ms
    Wire.begin();
    delay(100);

    // zero the encoders
    resetEncoders();

    // setup servo pointers
    Servo* servo_ptr;
    servo_ptr->attach(SERVOPIN);
    int ServoPosition = SERVOINIT;
    servo_ptr->write(ServoPosition);

    // initialise the course
    Course course = Course(servo_ptr, &ServoPosition);

    // wait a bit before we start
    delay(3000);

    // TESTING - loop through legs and run their action
    course.run();

    /*Line test = Line(500, FORWARD, 90, false);
    test.run();
    Circle testCircle = Circle(180, 270, BACKWARD, 90, true);
    testCircle.run();*/

    // turn on for event
    //finished();

}

// just a bit of fun
void finished() {
    tone(9,660,100);
    delay(150);
```

```
    tone(9,660,100);
    delay(300);
    tone(9,660,100);
    delay(300);
    tone(9,510,100);
    delay(100);
    tone(9,660,100);
    delay(300);
    tone(9,770,100);
    delay(550);
    tone(9,380,100);
    delay(575);
}

void loop() {

}
```

Course.cpp - Logic for completing the course

```
/*
 *  Course list - includes Legs variable that stores how to run the course
 */

#include "defines.h"
#include "Leg.cpp"
#include "Line.cpp"
#include "Circle.cpp"

// include guard
#ifndef COURSE_CPP
#define COURSE_CPP

class Course {
  public:
    Leg** legs;

    Course(Servo* servo_ptr, int* servoPosition_ptr) {
        legs = new Leg*[13];

        /*
        * ---- THE LEG CODE ----
        *
        * Each leg represents a part of the course, with the following parameters
        *      Line - Distance, Direction, Angle to turn at end, drop M&M
        *    Circle - Radius, angle to move through, direction, angle to turn at end, drop M&M
        */
        legs[0] = new Line(428, FORWARD, 0, true);
        legs[1] = new Line(504, FORWARD,-37, false);
        legs[2] = new Circle(180, 270, BACKWARD, 90, true);
        legs[3] = new Line(180, FORWARD, -40, false);
        legs[4] = new Line(622, BACKWARD,50, true);
        legs[5] = new Line(400, BACKWARD,-90, false);
        legs[6] = new Line(400, FORWARD, 90, true);
        legs[7] = new Line(400, FORWARD, 90, false);
        legs[8] = new Line(660, FORWARD, 90, true);
        legs[9] = new Circle(260, 90, FORWARD, -90, false);
        legs[10] = new Line(500, FORWARD, 90, false);
        legs[11] = new Line(260, FORWARD, 90, false);
        legs[12] = new Line(340, FORWARD, 143, false);

```

```
        for (int i = 0; i < 13; i++) {
            legs[i]->servo = servo_ptr;
            legs[i]->servoPosition = servoPosition_ptr;
        }
    }

    void run() {
        for (int i = 0; i < 13; i++) {
            Serial.print("Running leg ");
            Serial.println(i);

            legs[i]->run();
        }
    }
};

#endif
```

Leg.cpp - defines code for completing a section of the course

```
/*
 *  Leg class - controls the robot for one 'leg' (segment) of the course
 *              includes logic for the action, rotate, and stop functions
 */

#include "defines.h"
#include "encoderInteraction.cpp"

#include <Arduino.h>
#include <Servo.h>
#include <Wire.h>

// include guard
#ifndef LEG_CPP
#define LEG_CPP

class Leg
{
  public:
    // pointer to serveo position variable
    int* servoPosition;
    Servo* servo;

    // Should we drop an M&M?, leg finished successfully?
    bool drop, direction;

    // Virtual function that will be called to run this leg of the course.
    virtual void run();

    // perform actions at waypoint, including dropping M&M if needed
    void action() {
        // turn on LED and buzzer
        digitalWrite(LEDPIN, HIGH);
        tone(PIEZOPIN, PIEZOFREQ);

        // if need to drop M&M, drop one, if not delay so we can see and hear buzzer
        if (this->drop) { this->dispense(); }
        else { delay(NOTIFYPAUSE); }

        // turn off led & buzzer
        digitalWrite(LEDPIN, LOW);
```

```
42          noTone(PIEZOPIN);
43      }
44
45      // dispense an M&M
46      void dispense() {
47          // increase servo position
48          *servoPosition += SERVOSTEP;
49
50          // make sure we don't accidentally run through all positions
51          if (*servoPosition >= 179) {
52              *servoPosition = 179;
53          }
54
55          // write the servo position and wait to ensure clean drop
56          servo->write(*servoPosition);
57          delay(SERVOPAUSE);
58      }
59
60      // rotate by the given angle (+ve clockwise), returning the actual angle rotated
61      int rotate(int t, bool correction) {
62          Serial.print("Rotate ");
63
64          // find distance needed to rotate
65          int dist = (int)(2 * PI * WHEELDIST * ((float)abs(t) / (float)360));
66
67          Serial.print(dist);
68          Serial.print(" ");
69
70          // speeds of each wheel
71          int leftWheel, rightWheel, rotateSpeed;
72
73          // if we're in correction mode rotate slower
74          if (correction) {
75              rotateSpeed = DUALSPEED * 0.1;
76          } else {
77              rotateSpeed = DUALSPEED * 0.5;
78          }
79
80          Serial.print((int)rotateSpeed);
81
82          // set speeds of each wheel depending on direction (+ve -> left goes forwards)
83          if (t > 0) {
84              leftWheel = (128 + rotateSpeed);
85              rightWheel = (128 - rotateSpeed);
86          } else {
87              leftWheel = 128 - rotateSpeed;
88              rightWheel = 128 + rotateSpeed;
89          }
90
91          Serial.print(" ");
92          Serial.print((int)leftWheel);
93          Serial.print(" ");
94          Serial.println((int)rightWheel);
95
96          // while we've not rotated less that the required distance
97          while (averageDistance() <= dist) {
98              Serial.println(averageDistance());
99
100             // set wheels to spin at different speeds
101             Wire.beginTransmission(MD25ADDR);
102             Wire.write(MODE);
```

```
            Wire.write(MODESEPERATE);
            Wire.endTransmission();

            // set the acceleration mode to fast
            Wire.beginTransmission(MD25ADDR);
            Wire.write(ACCEL);
            Wire.write(ACCELDEFAULT);
            Wire.endTransmission();

            // Set left wheel speed
            Wire.beginTransmission(MD25ADDR);
            Wire.write(SPEEDLEFT);
            Wire.write((char)leftWheel);
            Wire.endTransmission();

            // set right wheel speed
            Wire.beginTransmission(MD25ADDR);
            Wire.write(SPEEDRIGHT);
            Wire.write((char)rightWheel);
            Wire.endTransmission();
        }

        int avg = averageDistance();

        // return the angle rotated
        int ang = (int)((float)(360 * avg)/(float)(2 * PI * WHEELDIST));

        resetEncoders();
        this->stop();
        delay(50);
        return ang;
    }

    // stop the vehicle
    void stop() {
        // allow both registers to be set to stop
        Wire.beginTransmission(MD25ADDR);
        Wire.write(MODE);
        Wire.write(MODESEPERATE);
        Wire.endTransmission();

        // high acceleration mode
        Wire.beginTransmission(MD25ADDR);
        Wire.write(ACCEL);
        Wire.write(10);
        Wire.endTransmission();

        // set left to stop
        Wire.beginTransmission(MD25ADDR);
        Wire.write(SPEEDLEFT);
        Wire.write(128);
        Wire.endTransmission();

        // set right to stop
        Wire.beginTransmission(MD25ADDR);
        Wire.write(SPEEDRIGHT);
        Wire.write(128);
        Wire.endTransmission();
        delay(50);
    }
};
```

```
164
165 #endif
```

Line.cpp - logic for driving a straight line

```
1  /*
2   *  Line class - drives a straight leg of the course
3   */
4
5  #include "defines.h"
6  #include "encoderInteraction.cpp"
7  #include "Leg.cpp"
8
9  #include <Arduino.h>
10 #include <Wire.h>
11
12 // include guard
13 #ifndef LINE_CPP
14 #define LINE_CPP
15
16 class Line: public Leg {
17     // count how many times we loop over the drive sections, so we don't get stuck.
18     int loopCount;
19   public:
20     // distance to travel, and how far to rotate to be pointing in correct direction at end of the leg
21     int dist, endRot;
22
23     // constructor
24     Line(int d, int dir, int r, bool m) {
25         this->dist = d;
26         this->direction = dir;
27         this->endRot = r;
28         this->drop = m;
29         this->loopCount = 0;
30     }
31
32     // implement the run function
33     void run() {
34
35         // run drive, get how far we actually drove
36         int driven = this->drive(this->direction * this->dist, false);
37
38         // calculate distance left to drive
39         int shortfall = this->dist - driven;
40
41         // aim to get within 2mm of the target waypoint, without going over MAXLOOPCOUNT
42         while (abs(shortfall) > LINEARTOL && loopCount < MAXLOOPCOUNT) {
43             Serial.print("Shortfall ");
44             Serial.println(shortfall);
45             // if we aren't on target, drive the shortfall again, looping over to check we reached it
46             driven = this->drive(shortfall, true);
47             this->stop();
48             shortfall = abs(shortfall) - driven;
49             this->loopCount++;
50
51         }
52         this->loopCount = 0;
53
54         // now repeat this for rotation
55         this->rotate(this->endRot, false);
56
```

```
        // blink light, sound buzzer, and drop M&M if needed
        this->action();
    }

    // move the wheels the desired distance, and return the actual distance driven
    int drive(int d, bool correction) {
        Serial.print("Line ");

        while(averageDistance() <= abs(d)) {
            int spd;

            // if in a correction, go slowly for more accuracy
            if (correction) {
                spd = DUALSPEED * 0.2;
            } else {
                spd = DUALSPEED;
            }

            // Set both wheels to spin at the same rate
            Wire.beginTransmission(MD25ADDR);
            Wire.write(MODE);
            Wire.write(MODEUNSIGNEDDUAL);
            Wire.endTransmission();

            // set the acceleration mode to fast
            Wire.beginTransmission(MD25ADDR);
            Wire.write(ACCEL);
            Wire.write(ACCELDEFAULT);
            Wire.endTransmission();

            // set the speed
            Wire.beginTransmission(MD25ADDR);
            Wire.write(SPEEDLEFT);

            // if we're given a negative distance, drive backwards
            if (d < 0) {
                Wire.write((char)(128 - spd));
            } else {
                Wire.write((char)(128 + spd));
            }
            Wire.endTransmission();
        }

        // return the read distance
        int avg = averageDistance();
        Serial.println(avg);
        resetEncoders();
        this->stop();
        delay(50);
        return avg;
    }
};

#endif
```

Circle.cpp - logic for driving an arc of the course

```
/*
 *  Circle class - contains logic for driving a circular section of the course
 */

```

```
5  #include "defines.h"
6  #include "encoderInteraction.cpp"
7  #include "Leg.cpp"
8
9  #include <Arduino.h>
10 #include <Wire.h>
11
12 // include guard
13 #ifndef CIRCLE_CPP
14 #define CIRCLE_CPP
15
16 class Circle: public Leg {
17     // loop counter to ensure we don't get stuck in a loop, distance the outer wheel has to rotate
18     int loopCount, direction, outerDist, innerDist;
19   public:
20     // radius of the circle, angular distance to travel, final rotation for next leg
21     int radius, theta, endRot;
22
23     // constructor
24     Circle(int r, int t, int dir, int eR, bool m) {
25         this->radius = r;
26         this->theta = t;
27         this->direction = dir;
28         this->endRot = eR;
29         this->drop = m;
30         this->loopCount = 0;
31
32         // compute the outerDist as 2*pi*(radius of circle + distance to outer wheel from center of rob
33         this->outerDist = (int)(2 * PI * (this->radius + WHEELDIST) * ((float)abs(this->theta) / 360));
34
35         // similar procedure for innerDist, but subtract the wheel distance instead
36         this->innerDist = (int)(2 * PI * (this->radius - WHEELDIST) * ((float)abs(this->theta) / 360));
37     }
38
39     // implement the run function
40     void run() {
41         Serial.print("Circle ");
42
43         Serial.print(this->radius);
44         Serial.print(" ");
45         Serial.print(this->outerDist);
46         Serial.print(" ");
47         Serial.print(this->innerDist);
48         Serial.print(" ");
49
50         // drive round in a circle
51         int driven = this->drive(this->radius, false);
52
53         // get angular shortfall
54         int angShortfall = this->theta - driven;
55
56         while (abs(angShortfall) <= ANGULARTOL && this->loopCount < MAXLOOPCOUNT) {
57             driven = this->drive(angShortfall, true);
58             angShortfall = abs(angShortfall) - driven;
59             this->loopCount++;
60         }
61         this->loopCount = 0;
62
63         // rotate to start of next leg
64         this->rotate(endRot, false);
65
```

```
66          // perform any actions needed
67          this->action();
68      }
69
70      int drive(int t, bool correction) {
71          char innerWheel, outerWheel, innerSpeed, outerSpeed;
72
73          // reset inner distance to theta
74          this->innerDist = (int)(2 * PI * (this->radius - WHEELDIST) * ((float)abs(t) / 360));
75
76          // if we're going forward, the left wheel is on the inside, else its the outside wheel
77          if (this->direction == FORWARD) {
78              innerWheel = ENCODELEFT;
79              outerWheel = ENCODERIGHT;
80              innerSpeed = SPEEDLEFT;
81              outerSpeed = SPEEDRIGHT;
82          } else {
83              innerWheel = ENCODERIGHT;
84              outerWheel = ENCODELEFT;
85              innerSpeed = SPEEDRIGHT;
86              outerSpeed = SPEEDLEFT;
87          }
88
89          // angular velocity from dual speed, with direction
90          float omega = this->direction * ((float)DUALSPEED * 0.5 / (float)this->radius);
91
92          // if correction, reduce the speed for greater accuracy
93          if (correction) {
94              omega *= 0.2;
95          }
96
97          Serial.print("Omega: ");
98          Serial.println(omega);
99
100         // loop through driving until one of the distances is over it's limit
101         while (individualDistance(innerWheel) <= this->innerDist) {
102             // Set wheels to spin at different rates
103             Wire.beginTransmission(MD25ADDR);
104             Wire.write(MODE);
105             Wire.write(MODESEPERATE);
106             Wire.endTransmission();
107
108             // set the acceleration mode to fast
109             Wire.beginTransmission(MD25ADDR);
110             Wire.write(ACCEL);
111             Wire.write(ACCELDEFAULT);
112             Wire.endTransmission();
113
114             // Set outer wheel speed
115             Wire.beginTransmission(MD25ADDR);
116             Wire.write(outerSpeed);
117             Wire.write((char)(128 + (this->radius + WHEELDIST) * omega));
118             Wire.endTransmission();
119
120             // set inner wheel speed
121             Wire.beginTransmission(MD25ADDR);
122             Wire.write(innerSpeed);
123             Wire.write((char)(128 + (this->radius - WHEELDIST) * omega));
124             Wire.endTransmission();
125         }
126
```

```
        int innerDriven = individualDistance(innerWheel);
        // get the angle driven through
        int ang = 360 * innerDriven / (2 * PI * (this->radius - WHEELDIST));
        resetEncoders();
        this->stop();
        return ang;
    }

};

#endif
```

encoderInteraction.cpp - functions for interacting with the MD25 encoders

```
/*
 *  Encoder interaction file, contains functions to read and clear MD25 encoders
 *          Uses inline functions to prevent multiple definitions
 */

#include "defines.h"

#include <Arduino.h>
#include <Wire.h>

// include guard
#ifndef ENCODERINTERACTION_CPP
#define ENCODERINTERACTION_CPP

// find distance a specific wheel has moved
inline int individualDistance(char side) {
    // set MD25 to send the encoder for the given side
    Wire.beginTransmission(MD25ADDR);
    Wire.write(side);
    Wire.endTransmission();

    // request 4 bytes from the MD25
    Wire.requestFrom(MD25ADDR, 4);

    // wait for first 4 bytes back
    while (Wire.available() < 4);

    // get all bytes of the click var
    long clicks = Wire.read();
    clicks <<= 8;
    clicks += Wire.read();
    clicks <<= 8;
    clicks += Wire.read();
    clicks <<= 8;
    clicks += Wire.read();

    delay(5);

    // convert clicks to mm
    int dist = clicks * CLICKSTOMM;

    // return absolute distance moved
    return abs(dist);
}

// reset distance encoders between legs
inline void resetEncoders() {
```

```
    Wire.beginTransmission(MD25ADDR);
    Wire.write(CMD);
    Wire.write(CLEARENCODERREGISTERS);
    Wire.endTransmission();
    delay(50);
}

// find the average distance travelled
inline int averageDistance() {
    // get individual wheel distances
    int distLeft = individualDistance(ENCODELEFT);
    int distRight = individualDistance(ENCODERIGHT);

    // find the absolute distance
    distLeft = abs(distLeft);
    distRight = abs(distRight);

    // return the average
    return (int)((distLeft + distRight)/ 2);
}

#endif
```

defines.h - header including all definitions

```
/*
 *  Defines for odometry task
 */

// include guard
#ifndef DEFINES_H
#define DEFINES_H

// constant definitions
#define MAXLOOPCOUNT 5          // maximum times to loop while correcting steer/drive
#define WHEELDIST 125           // distance between centre of robot and centre of wheels
#define PIEZOFREQ 1000          // frequency to sound the buzzer at
#define NOTIFYPAUSE 200         // time to sound buzzer and flash light if not dropping M&M
#define SERVOINIT 0             // initial angle for servo to sit at (the empty hole)
#define SERVOSTEP 34            // angle to rotate servo by in order to move to next hole
#define SERVOPAUSE 400          // time to wait to ensure M&M drops cleanly
#define DUALSPEED 50            // speed of the motors in dual mode
#define FORWARD 1               // multiplier to move forward
#define BACKWARD -1             // backwards multiplier
#define LINEARTOL 2             // linear tolerance for accuracy in straight line
#define ANGULARTOL 2
#define CLICKSTOMM 0.890        // conversion factor from clicks to mm

// MD25 I2C codes
#define MD25ADDR 0x58           // I2C MD25 address
#define SPEEDLEFT 0x00          // MD25 register for speed #1 (left)
#define SPEEDRIGHT 0x01         //   "      "       "      #2 (right)
#define ENCODELEFT 0x02         // encoder address left
#define ENCODERIGHT 0x06        // "        "      right
#define ACCEL 0x0E              // Acceleration encoder
#define MODE 0x0F               // mode register
#define CMD 0x10                // command register

// MD25 command codes
#define CLEARENCODERREGISTERS 0x20  // code to clear encoder values

```

```
37  // MD25 acceleration modes
38  #define ACCELDEFAULT 2          // acceleration mode
39
40  // MD25 modes
41  #define MODEUNSIGNEDDUAL 2
42  #define MODEDUAL 3              // dual motor mode, all off speed 1
43  #define MODESEPERATE 0          // seperate motor speeds
44
45  // pin definitions
46  #define LEDPIN 8                // led pin
47  #define PIEZOPIN 9              // buzzer spin
48  #define SERVOPIN 10             // servo pin
49
50  #endif
```